
1   [PROB]
2   # Population pharmacokinetics of tacrolimus
3   woillard de winter bjcp 2011
4   
5   [SET] end=10*12, delta=0.1
6   
7   [PARAM] @annotated
8   TVCL : 21.2 : Typical value of clearance (L/h)
9   TVV1 : 486 : Typical central volume of distribution (L)

10   TVQ : 79 : Typical intercomp clearance 1 (L/h)
11   TVV2 : 271 : Typical peripheral volume of distribution 1 (L)
12   TVKTR : 3.34 : Typical transfert rate constant(1/h)
13   HTCL : -1.14 : effect of HT on cl
14   CYPCL : 2.00 : effect of CYP on cl
15   STKTR : 1.53 : effect of study on ktr
16   STV1 : 0.29 : effect of study on v1
17   HT : 70 : hematocrit (perc)
18   ST : 1 : Prograf(1) adv(0)
19   CYP : 0 : expressor (1) non exp(0)
20   
21   
22   [CMT] @annotated
23   DEPOT : Dosing compartment (mg)
24   TRANS1 : transit compartment 1 (mg)
25   TRANS2 : transit compartment 2 (mg)
26   TRANS3 : transit compartment 3 (mg)
27   CENT : Central comaprtment (mg)
28   PERI : First peripheral compartment (mg)
29   AUC : AUC cp
30   
31   
32   [OMEGA] @annotated
33   ETACL : 0.28 : ETA on clearance
34   ETAV1 : 0.31 : ETA on V1
35   ETAQ : 0.54 : ETA on Q
36   ETAV2 : 0.60 : ETA on V2
37   ETAKTR : 0.24 : ETA on KTR
38   
39   [MAIN]
40   double CL = TVCL*pow(HT/35,HTCL)*pow(CYPCL,CYP)*exp(ETACL);
41   double V1 = TVV1*pow(STV1,ST)*exp(ETAV1);
42   double Q = TVQ*exp(ETAQ);
43   double V2 = TVV2*exp(ETAV2);
44   double KTR = TVKTR*pow(STKTR,ST)*exp(ETAKTR);
45   
46   [SIGMA] @annotated
47   ADD : 0.78 : additive Residual unexplained variability
48   PROP : 0.113: proportionnal Residual unexplained variability
49   
50   [ODE]
51   dxdt_DEPOT = -KTR*DEPOT;
52   dxdt_TRANS1 = KTR*DEPOT -KTR*TRANS1;
53   dxdt_TRANS2 = KTR*TRANS1 -KTR*TRANS2;
54   dxdt_TRANS3 = KTR*TRANS2 -KTR*TRANS3;
55   dxdt_CENT = KTR*TRANS3 -(CL+Q)*CENT/V1 + Q*PERI/V2 ;
56   dxdt_PERI = Q*CENT/V1 - Q*PERI/V2 ;
57   dxdt_AUC = CENT/(V1/1000);
58   if(SS_ADVANCE) dxdt_AUC = 0;
59   
60   [TABLE]
61   capture CP = (CENT/(V1/1000))*(1+PROP) + ADD;
62   

